This is output from your simulation without any changes.

First look a physical cross section.

(1) The spacers are far too wide turning your 9.5nm transistor into a 29.8nm transistor. That is like stepping back 5 generations of transistor. So the spacer has to be changed.

(2) The profile of the poly gate is not good. This is due to the wet oxidation done after the poly is etched. This would never be done on even 0.25um technologies so this has to change.

(3) Look at the doping beneath the poly (the channel of the transistor. It looks like the source and drain implants are shorted beneath the poly. The cut line graph of doping confirms there is boron in the channel. This is because you use bf2 for the threshold adjust implant.



Bf2 is a threshold adjust implant for an nmos transistor not a pmos. Pmos would use phosphorous.

So lets step through your process and make necessary changes……..

From start though to end of Well formation there are a few comments but nothing needs to be changed.

```
# Start the Athena Process simulator running...
go athena
#
# Set up a mesh suitable for a single MOSFET device....
# Set up a mesh suitable for a single MOSFET device....
line x loc=0 spac=0.1
line x loc=0.35 spac=0.02
line x loc=0.5 spac=0.1
#
line y loc=0.00 spac=0.005
line y loc=0.3 spac=0.015
line y loc=0.5 spac=0.02
line y loc=2 spac=0.2
line y loc=5 spac=1
#
# Start off by defining silicon with 1e14 phos doping...
# Decrease the folowing space.mult parameter for a denser
# mesh and more accuracy...
init orientation=100 c.phos=1e15 space.mult=2
#
#pwell formation including masking off of the nwell
#
#---This next step is redundant as it does not change anything. It was probably
#---the anneal step for the PWell formation
diffus time=30 temp=1000 dryo2 press=1.00 hcl=3
#---This next step removes oxide grown by the previous step. Both can be removed.
etch oxide thick=0.02
#
#N-well Implant
implant amorphous phos dose=1.0e12 energy=100 pears
#
diffus temp=950 time=100 weto2 hcl=3
#
#N-well implant amorphous not shown ---Yes it was !!! That is what previous step is.
#
# welldrive
diffus time=220 temp=1200 nitro press=1
#---The next step is redundant as the above well drive does not use an
#---oxidising step. It does not harm anything but could lead to confusion.
#---Lets leave it for now.
etch oxide all
```

Next is the gate oxide module.
The sacrificial oxide is a means to cleanup the wafer surface and could be removed to say simulation time. But read the comments about the change of doping at the wafer surface due to loss of silicon during oxidation.
Lets leave it in place.

Now for gate oxidation you used temperatures of <450ºC. Silicon will not oxidise at temperatures below 800ºC – just the laws of physics. One exception is that a sort of monolayer of oxide will form on a silicon surface giving an oxide thickness of between 1 and 2nm.
Silvaco assumes this so it can appear that you have grown oxide when it was there all along.

We must change the gate oxidation process, there is no other way to do this.
Changes:
- Steps added are in **BLUE**
- Steps deleted are in **GREEN**
- Comments added are in **RED**

```
#sacrificial "cleaning" oxide
#---This is a "trick" used by modern Wafer Fabs to remove defects that may have been
#---present on the wafer surface either from the starting silicon, resist residues
#---from photo patterning (carbon will be burned off at 1000C) and any heavy metals
#---coming from the implanters. I(t also helps anneal the surfaceof the silicon
#---crystal lattice. BUT !! it also removes dopant from the silicon surface as some
#---silicon is consumed growing the oxide. So your implants must accomodate for
#---this loss. Not important for this simulation but always remember this in future.
diffus time=20 temp=1000 dryo2 press=1 hcl=3
#
etch oxide all
#
#gate oxide grown here:-
#---This cannot work. Silicon dioxide will not grow beklow about 800 degrees C!
#---Need to completely change this.
#set partial_press=1.0
#diffus time=9 temp=350 nitrogen hcl=1
#diffus time=2 temp=450 dryo2 press=$partial_press
#diffus time=11 temp=422 dryo2 press=$partial_press hcl=1
#diffus time=8 temp=450 nitrogen hcl=1
#diffus time=8 temp=350 t.final=750 nit hcl=1
#
#---Now the default for partial pressure is 1.0 so this step is redundant
#---but harmless
set partial_press=1.0
#---Next, Silvaco assumes that any exposed silicon already has a native oxide between
#---1 and 2nm. This is actually true. When a wafer is sent for gate oxidation it is
#---first cleaned in dilute HF to remove any surface oxide, then quickly moved to
#---the gate oxidation step. So if you want a gate oxide <2nm, you have to tell
#---the simulator to switch this off. This is the next statement:
oxide init=0.0001
#---Now a gate oxide of <2.5nm cannot use a furnace process because the time required
#---to grow the thin oxide is too short. What is used is a single wafer rapid thermal
#---processor -which is a glorified box with halogen lamps in it.
#---This can heat the wafer to 1000C in seconds! The next step simulates this.
diffuse temp=1000 seconds time=26 dryo2
# Extract a design parameter.....
extract name="gateox" thickness oxide mat.occno=1 x.val=0.005
#
```

Next is the poly gate module.

First the VT adjust implant is destined to go under the poly gate in the channel and adjusts the threshold voltage. In a modern process, this is incorporated into the well implant at the start of the process so you are not implanting through the gate oxide which needs to be as near perfect as possible.
This is why my code had two implants for the NWell, the one closest to the surface is the threshold adjust. (A real process will have 3 implants).

Now your code uses BF2 which implants boron which is the same species as the source drain implant. So In fact this will short out your transistor and give you a zero threshold voltage. We have to change this to phosphorous. Use these conditions to change the VT.

Next is the poly oxidation step. Again this is for very old technology and cannot be used when gate lengths are narrow as the oxidation consumes too much of the polysilicon.
We have to remove this step.

Next you are using an LDD implant. This is not required for pmos transistors as boron moves quite far under subsequent thermal processes. If you leave it in place, the Leff of the transistor will be very small or may even short out the transistor just like the wrong VT adjust implant. We must remove this step.

```
#vt adjust implant amorphous
#---This is wrong. This implants boron everywhere so under the pmos channel
#---which will short out the transistor. This is an nmos VT adjust implant
#---so must be changed
#implant amorphous bf2 dose=1.5e12 energy=10 pearson
#---try this
implant phos dose=1.5E13 energy=32 tilt=5 pearson
#
depo poly thick=0.250 div=3
#
etch poly left p1.x=0.45
#
# Relax the mesh below the 0.5um plane, for speed....
relax y.min=0.5
#
#---The next two steps oxidise exposed silicon and poly silicon.
#---This is used in older technologies where this oxide growth is a small
#---percentage of the poly thickness. It would never be used below a 180nm
#---process. It must be removed.
#method fermi compress
#diffuse time=5 temp=900 weto2 press=1
#
# PLDD implant amorphous
#---Again this is for older technologies where spacers were significantly bigger
#---than for 90nm process. They are used for nmos transistors but not pmos as boron
#---will diffuse beneath the spacer to rech the channel.
#---So we must remove this step or the Leff will be too low and the transistor will
#---leak like hell.
#implant amorphous bf2 dose=6e15 energy=20 pearson
#
```

Last is the spacer module

Your code forms a huge spacer of nearly 100nm on each side. This is from a 0.25um or greater process and not for 90nm.
So we need to change this.

We need to make the P+ implant a little shallower for 90nm so back off the energy from 20 to 15keV.

And finally, we cannot use a furnace anneal to activate the P+ implant as it will diffuse too far into the channel and cause too small an Leff. You must use a spike anneal – no choice here.

```
# This is a good way of defining the spacer.....define a variable
# first with the 'set' command....
#---The spacer should be about 10% the total width of the minimum poly gate.
#---so in this case <9nm. So change next step from
#set spacer=0.10
#---to
set spacer=0.015
depo oxide thick=$"spacer" divisions=5
etch oxide dry thick=$"spacer"+0.004
#
# P++ Implant
implant amorphous bf2 dose=1.0e15 energy=20 pearson
# Final anneal.
#---Cannot use furnace anneal as boron will laterally diffuse too far.
#---Must use spike anneall which uses a special rapid thermal processor.
#method fermi compress
#diffuse time=5 temp=900 nitro press=1.0
diffuse time=3 seconds temp=1000 nitro press=1.0
#
```



Now you must admit this looks better. Gate oxide is 15 Angs, VT is 185mV

Further improvements

Look at how ragged the contours are in the channel here.
This is because the mesh is not very good.



Also look at the wasted space either side of the gate poly.

We only need about 0.2um either side of the poly edge.



Changing the as follows:

line x loc=0.00 spac=0.1
line x loc=0.10 spac=0.005
line x loc=0.20 spac=0.005
line x loc=0.45 spac=0.01
#
line y loc=0.00 spac=0.005
line y loc=0.1 spac=0.015
line y loc=0.5 spac=0.02
line y loc=2 spac=0.2

This also requires poly etch to be changed

etch poly left p1.x=0.2

And the contact metal

where electrodes are required.....
etch oxide left p1.x=0.1
deposit alumin thick=0.015 div=2
etch alumin right p1.x=0.1

Ah… much more beautiful and more realistic. But it is now showing up a real problem with this process. Look at the contours inside the poly. There is no doping near the poly/gate oxide interface. So the poly will be undoped here and will act as an insulator. This did not show up on the previous code because the mesh was not accurate enough.

The graph on the right shows the boron profile (RED) in the poly drops to very low close to the surface of the wafer!!
This device will not work because the 90nm P+ must be a shallow implant, but it is too shallow to dope the poly.



The only way to fix this is to dope the poly after it is deposited before etch.

```
depo poly thick=0.200 div=3
implant amorphous boron dose=1.0e15 energy=25 pearson
#
etch poly left p1.x=0.2
```

Now see the difference: The poly is fully doped



So this is more accurate, but there is a consequence of this increased accuracy. Without changing anything else, the VT has increased to almost 200mV. So the VT adjust implant will need to be fine tuned.

So although we can make your code work, it does not really represent reality. For 90nm or below, the poly requires to be doped after deposition and the VT adjust implant is part of the well implant.